Differential Amplifiers

- The differential amplifier (or diff-amp) is a foundational building block for most analog ICs today. They may be constructed with BJTs, MOSFETS, or even tubes!
- Diff-amps have been around since the 1940's and are still very widely used today.
- Previously, we looked at amplifiers that amplified the voltage between a *single input* and ground.
- The diff-amp amplifies the *difference* between *two voltages*. This is done largely irrespective of their potential to ground. The diff-amp output is the difference between the two inputs multiplied by the differential gain A_d of the amplifier.

Differential Amplifiers

- Let's consider one form the BJT diff-amp can take as shown below. This is also called a *long-tailed* or *emitter-coupled* pair.



Figure: BJT Differential Amplifier

- The BJTs are identical pairs with their bases DC biased at the same voltage. The emitters are connected to a large negative voltage through a common large resistance mimicking a current source. The collectors have identical resistors connected to V_{cc}.

Differential Amplifiers

- The diff-amp can have differential inputs and outputs, differential inputs and a single-ended output, single-ended input and differential outputs or single-ended input and output.
- This makes for a very versatile amplifier building block.



Differential Amplifiers

- In *common mode*, the same input signal is applied to both V_{b1} and V_{b2}. In this case, the circuit acts as if it has only one transistor with R_{c1} and R_{c2} in parallel. The voltage gain in this case would be quite small due to large R_e.
- Low common-mode gain makes the diff-amp excel in distinguishing differential signals from common-mode noise.



Differential Amplifiers

- In *differential mode*, the input signal is applied differentially to V_{b1} and V_{b2}. In other words, V_{b1} goes up and V_{b2} goes down. Since the summation of the emitter currents is fixed by the constant current source, V_{c1} goes down and V_{c2} goes up.



Differential Amplifiers

- Common mode signals are common to each input. A diff-amp rejects common mode and amplifies the differential mode signals.



Figure: Differential signals with a Common Mode Signal



Figure: Diff-amps are used in RS-422 signaling

TI App. Report: slla070d.pdf

Differential Amplifiers

- Let's analyze a diff-amp. One of the most common configurations for the diff-amp is where the input is differential and the output is single-ended. This configuration is shown below.
- Note that V_{b1} and V_{b2} are DC biased at zero volts and that the differential input signal is applied as two sources with one 180° out of phase with the other.



Differential Amplifiers

- First, determine the DC steady state of the amplifier.

We can determine the quiescent emitter current by forcing the input signals to be at zero volts and writing the KVL loop around the base emitter junction starting at ground, just below V2, remembering that Q2 supplies only half the current through R3.



$$-0 + 0.6 + \frac{I_{R3}(7200)}{2} - 15 = 0$$

$$\frac{I_{R3}(7200)}{2} = 15 - 0.6$$

$$\frac{I_{R3}}{2} = \frac{15 - 0.6}{7200}$$

$$I_{R3} = 1mA; \textit{ Q2's emitter is half of R3's current}$$

Knowing I_e we can find g_m:

$g_m = \frac{I_e}{V_t} = \frac{.001}{.026} = 0.03845S$

Differential Amplifiers

- We know the small signal voltage at the collector of a common-emitter amplifier is: $v_c = g_m V_{be} R_c$.
- The gain for this circuit will be $\frac{v_{out}}{v_{in}}$ or $\frac{v_{c2}}{v_{diff}}$. This is the single-ended output voltage divided by the differential input voltage.
- The differential voltage is split evenly between the B-E junctions. Thus, the voltage at the emitter's junction can't change. If so, that junction is a *virtual* or AC ground. Following that reasoning, we see that voltage across the B-E junction v_{be} at each transistor is $0.5 v_{diff}$.



Figure: Each B-E junction sees one-half of v_{diff}

Differential Amplifiers

- Now we can compute the small signal voltage at the collector of Q2 remembering that $V_{diff} = 2mV$.

$$v_{be} = \frac{v_{diff}}{2}; so,$$
$$v_c = (\frac{v_{diff}}{2})g_m R_c$$
$$= .001(.03845)(1000) = 0.03845$$

- From before, the gain for this circuit will be $\frac{v_{out}}{v_{in}}$ or $\frac{v_{c2}}{v_{diff}}$.
- Our single-ended, small-signal v_c is 0.03845. What is the differential input voltage? It's just twice the single-ended input signal or simply 2mV.

$$A_v = \frac{v_{out}}{v_{in}}$$
$$= \frac{v_{c2}}{v_{diff}} = \frac{0.03845}{.002} = 19.23$$

Differential Amplifiers

- Time to spin-up the simulator!




```
.title adi.sp

.include 2n4401.mod

Vin1 q1_b gnd sin(0 0.5m 1000 0 0   0) ;   +0 deg
Vin2 q2_b gnd sin(0 0.5m 1000 0 0 180) ; +180 deg

Vcc    vcc    gnd             15v
Vee    vee    gnd            -15V

Q1     q1_c   q1_b   q1_e   2n4401 ;emitters tied
Q2     q2_c   q2_b   q1_e   2n4401

R3     q1_e    vee            7200

RC1    vcc    q1_c            1.0K
RC2    vcc    q2_c            1.0K

C1     q2_c   out             10u ;coupling cap
R4     out    gnd             1G  ;dc path to gnd

.control
  destroy all
  set hcopydevtype=postscript
  set hcopypscolor=true
  set color0=rgb:f/f/f
  set color1=rgb:0/0/0
  op                 ;analyze DC operating point
  tran 20u 5m        ;plot every 20uS for 5mS
  plot  v(q1_b,q2_b) v(out) ; plot AC voltages
.endc
.end
```

Differential Amplifiers

- Output from the simulation, differential input, single-ended output. Gain is about 19.



Differential Amplifiers

- What about the common mode gain? This is where both inputs to the diff-amp come from the same source.




```
.title adi_cm.sp
*diff-amp but common-mode signal
.include 2n4401.mod

Vin1 q1_b gnd sin(0 0.5m 1000 0 0   0) ;   +0 deg

Vcc    vcc    gnd              15v
Vee    vee    gnd              -15V

Q1    q1_c   q1_b   q1_e   2n4401 ;emitters tied
Q2    q2_c   q1_b   q1_e   2n4401 ;bases tied too!

R3    q1_e    vee              7200

RC1    vcc   q1_c              1.0K
RC2    vcc   q2_c              1.0K

C1     q2_c   out              10u ;coupling cap
R4     out    gnd              1G  ;dc path to gnd

.control
  destroy all
  set hcopydevtype=postscript
  set hcopypscolor=true
  set color0=rgb:f/f/f
  set color1=rgb:0/0/0
  op                 : analyze DC operating point
  tran 20u 5m        ;plot every 20uS for 5mS
  plot  v(q1_b) v(out) ; plot AC voltages
.endc
.end
```

Differential Amplifiers

- Output from the simulation: common mode input (1mV), single-ended output (69uV). Gain is about .07!



Differential Amplifiers

- If we use a current source instead of the 7.2K emitter resistor to *vee*, then the common mode gain drops considerably. The effective shared R_e becomes very large as its replaced with a current sink.



Differential Amplifiers

- $A_{cm} = \frac{22nV}{1mv} = .000022!$